

\$EOLCOM //
\$TITLE Afghan Water Rights Priority Model
\$OFFSYMREF OFFSYMLIST OffLISTING OFFUPPER

OPTION LIMROW=000, LIMCOL = 0;

\$ONTEXT

EOLCOM // tells GAMS to ignore anything in the line's text after the symbol //
OffLISTING deletes all program lines and just includes GAMS listing
Set LIMROW = 0 to eliminate all equations in the GAMS listing
Set LIMROW = 100 or more to show all equations in listing. Helps trap errors

* -----

*Frank A. Ward:
Dept of Agr Economics/Agr Business
New Mexico State University, Las Cruces, NM USA
e-mail: fward@nmsu.edu*

*Dina A. Salman:
Dept of Agr Economics/Agr Business
New Mexico State University, Las Cruces, New Mexico USA
email: dinasalm@nmsu.edu*

*Saud A.Amer:
US Geological Survey - International Water Resources Division
Reston VA USA
email: samer@usgs.gov*

May 13, 2013

* -----

* *Model 4 Simple Farm management*
* *Adds a SET (r) for water-sharing rule based on priority allocation of water as parameter*
* *water sharing rule requires introducing a priority SET (j)*
* *Also solves several models automatically inside a series of loops*
* -----

\$OFFTEXT

***** Section 1 *****
* Sets *

SETS

i province

/
1-Mousil
2-Basra
/

k crop

/ 1-wheat
2-cotton
/

s hydrologic water supply scenario

/
normal
dry
/

j water right priority: same number # of elements as # of provinces - 1 priority per province

/ j1*j2
/

r water allocation rule label: allows evaluating many water shortage sharing rules

/
us_priority upstream priority //rule 1
prop_sharing proportional sharing of shortages //rule 2
/

set rji(r,j,i) mapping set: assigns water use priorities to province - separate mapping for each proposed rule

/ us_priority . (j1.1-Mousil, j2.2-Basra)
prop_sharing . j1. (1-Mousil, 2-Basra)
/

display rji;

***** Section 2

* Data

*

table Bc(i,k) crop water demand(ET) (1000s cubic meters per Ha = 10ths of meters depth)

	1-wheat	2-cotton
1-mousil	11.9	18.0
2-Basra	13.5	21.4

*Bc(k) = 0.5 * Bc(k)

table Yield_p(i,k) Crop Yield tons per Ha (proportional to ET)

	1-wheat	2-Cotton
1-Mousil	1.39	2.32
2-Basra	1.57	2.75

* economic data

parameter Price_p(k) Crop Prices (\$ US per ton)

/1-Wheat	225
2-Cotton	906
/	

table Cost_p(i, k) Crop Production Costs excluding water (\$ US per Ha)

	1-wheat	2-cotton
1-Mousil	200	1286
2-Basra	207	1345

;

* data from Dr. Saud Amer 2012

Table land_p(i,k) observed land in prodn year 2006 (1000 Ha)

	1-wheat	2-cotton
1-mousil	47.4	0.45
2-basra	16.9	0.00

Parameter Net_revenue_p(i,k) calculated net revenue per unit land (\$ US per HA)

Net_revenue_p(i,k) = Price_p(k) * Yield_p(i,k) - Cost_p(i,k);

parameter Wat_supply_p(s) total water available (million cubic meters per year)

Wat_supply_p('normal') = sum((i,k), land_p(i,k) * Bc(i,k)); // calculates observed water use
Wat_supply_p('dry') = 0.5 * wat_supply_p('normal'); // dry: half of observed water use

parameter right_p(i) total water assigned by province in full supply conditions (paper water)

right_p(i) = sum(k, Bc(i,k) * land_p(i,k)); // sums over crops: ith province's historical (2006) water use under full supply

* code below loops over each priority (j) then maps each priority to corresponding province i
* code is thanks to Pete Stacy at GAMS Development Corporation Feb 1 2012

parameter tot_assigned(r,s) cumulative water assigned to province (i) including higher priorities;
tot_assigned(r,s) = 0; // starts at 0 - prepares to loop

parameter

tot_pap_right_by_prior(r, j) cumulative paper right at jth priority for all provinces including ties
remain_supply_by_prior(r,s,j) residual supply at jth priority after supplying higher priorities
wet_wat_use (r,s,i) wet (not paper) water use assigned to ith province - limited by total basin supply

Loop(r, // loop over sharing rule (r)
Loop(j, // loop over priority (j)

tot_pap_right_by_prior(r,j) = sum[i\$rji(r, j, i), right_p(i)] ; // total paper rights by (j) priority by provinces includes ties

remain_supply_by_prior(r,s,j) =

```

        min[(wat_supply_p(s) - tot_assigned(r,s)), tot_pap_right_by_prior(r,j)]; // remaining supply by (j) priority
after protecting higher priorities
    Loop(i$rji(r,j,i),
        wet_wat_use(r,s,i) = (right_p(i)/tot_pap_right_by_prior(r,j))
        * remain_supply_by_prior(r,s,j) + eps; // wet water assigned to (i) province
        tot_assigned(r,s) = tot_assigned(r,s) + wet_wat_use(r,s,i) ; // cumulative water (check) assigned to
last province getting water - should match total supply

    ); // end province loop
); // end priority loop
); // end rule loop

```

```

parameter tot_wat_use(r,s) total water use defined;
tot_wat_use (r,s) = sum(i, wet_wat_use(r,s,i)) + eps; // checks total use assigned against supply

```

***** end of water allocation system *****

**** begin looping over water sharing rule and water supply scenario below
* rr, ss, are subset of original sets... allows fast scenario analysis below

```

set rr(r); // water sharing rule
set ss(s); // water supply scenario

```

```

rr(r)=no; // switches subset off for now
ss(s)=no; // switches subset off for now

```

```

***** SECTION 3 *****
*                               Variables                               *
*****

```

positive variables

hectares_v	(r,i,k,s)	land in production by rule-province-crop-scen	(1000 Ha - marginal is \$US per Ha)
T_hectares_v	(r,i, s)	total land in prodn by rule-province-scen	(1000 Ha - marginal is \$US per Ha)
uses_crop_v	(r,i,k,s)	total water use by rule-province-crop-scen	(million m ³ - marginal is \$US per 1000 m ³)
uses_v	(r,i, s)	total water use by rule-province-scen	(million m ³ - marginal is \$US per 1000 m ³)

variables

ag_ben_k_v	(r,i,k,s)	total farm income by rule-province-crop-scen	(\$US 1000s - no marginals shown)
ag_ben_v	(r, s)	total farm income by rule-scen	(\$US 1000s - no marginals shown)

Tot_b_v total farm income re-calc for each rule-scen (objective) (\$US 1000s - no marginals shown)

```
***** Section 4 *****  
* Equations *  
*****
```

Equations

// Equations DECLARED

```
T_hectares_e    (r,i, s)    total land in production by province-scen    (1000 Ha - marginal is $US per Ha)  
Uses_crop_e    (r,i,k,s)    total water use by crop-province-scen    (million m^3 - marginal is $US per 1000 m^3)  
uses_e        (r,i, s)    total water use by scen                    (million m^3 - marginal is $US per 1000 m^3)  
  
ag_ben_k_e     (r,i,k,s)    total farm income by crop-province-scen    ($US 1000s - no marginals shown)  
ag_ben_e       (r, s)      total farm income by scen                    ($US 1000s - no marginals shown)  
  
Tot_b_e                    total farm income by element of loop        ($US 1000s - no marginals shown)
```

;

// Equations defined (using above names with algebraic formulas)

// Equations below defined over rule (rr) and scenario (ss) using subset rr and ss

```
T_hectares_e    (rr,i, ss).. T_hectares_v(rr,i, ss)    =e= sum(k, hectares_v (rr,i,k,ss));  
Uses_crop_e    (rr,i,k,ss).. uses_crop_v (rr,i,k,ss)    =e= Bc(i,k) * hectares_v(rr,i,k,ss);  
uses_e        (rr,i, ss).. uses_v        (rr,i, ss)    =e= sum(k, uses_crop_v(rr,i,k,ss));  
  
ag_ben_k_e     (rr,i,k,ss).. ag_ben_k_v (rr,i,k,ss)    =e= net_revenue_p(i,k) * hectares_v(rr,i,k,ss);  
ag_ben_e       (rr, ss).. ag_ben_v       (rr, ss)      =e= sum((i,k), ag_ben_k_v(rr,i,k,ss));  
  
Tot_b_e                    .. Tot_b_v                    =e= sum((rr,ss), ag_ben_v(rr,ss));    // sums total benefits  
over indices for sensitivity analysis
```

```
***** Section 5 *****  
* Labels and defines models *  
* Each model has one objective function *  
*****
```

model TE_04 /all/ ;

```

***** Section 6 *****
*                               BOUNDS                               *
*   Bounding shows a positive shadow price for each limiting resource   *
*****

uses_v.up   (r,i, s) = wet_wat_use(r,s,i ); // upper bound limits available water to observed water use
hectares_v.up(r,i,k,s) = land_p( i,k); // can produce no more land by crop than under full water supply

*hectares_v.lo('1-mousil','1-wheat') = 10; // food security requirement
*uses_crop_v.lo('wheat') = 1; // another way of enforcing food security

***** Section 7 *****
*                               SOLVE THE MODEL                               *
*****

loop(r,
loop(s,

    ss(s) = yes;
    rr(r) = yes;

    Solve TE_04 using nlp maximizing Tot_b_v; // multiple solves inside loop avoids writing solve several times

* closes loops below over r,s (rule for sharing shortages, water supply scenario)

    rr(r) = no;
    ss(s) = no;

);
);

// THE END

```